
Abstract 

 
Complex, end-to-end network services are set up via the configuration method: each component has a finite 
number of configuration parameters each of which is set to a definite value.  End-to-end network service 
requirements can be on connectivity, security, performance and fault-tolerance. However, there is a large 
conceptual gap between end-to-end requirements and detailed component configurations. To bridge this 
gap, a number of subsidiary requirements are created that constrain, for example, the protocols to be used, 
and the logical structures and associated policies to be set up at different protocol layers. By performing 
different types of reasoning with these requirements, different configuration tasks are accomplished. These 
include configuration synthesis, configuration error diagnosis, configuration error fixing, reconfiguration as 
requirements or components are added and deleted, and requirement verification. However, such reasoning 
is currently ad hoc. Network requirements are not even precisely specified hence automation of reasoning 
is impossible.  This is a major reason for the high cost of network management and total cost of ownership.  
 
This paper shows how to formalize and automate such reasoning using a new logical system called Alloy. 
Alloy is based on the concept of model finding. Given a first-order logic formula and a domain of 
interpretation, Alloy tries to find whether the formula is satisfiable in that domain, i.e., whether it has a 
model. Alloy is used to build a Requirement Solver that takes as input a set of network components and 
requirements upon their configurations and determines component configurations satisfying those 
requirements. This Solver is used in different ways to accomplish the above reasoning tasks. The Solver is 
illustrated in depth by carrying out a variety of these tasks in the context of a realistic fault-tolerant virtual 
private network with remote access. Alloy uses modern satisfiability solvers that solve millions of 
constraints in millions of variables in seconds. However, poor requirements can easily nullify such speeds. 
The paper outlines approaches for writing efficient requirements. Finally, it outlines directions for future 
research.  
 

                                                 
1 © 2004 Telcordia Technologies, Inc. 

1. Introduction 
 
Complex, end-to-end network services are set up via the configuration method: each component has a finite 
number of configuration parameters each of which is set to a definite value.  End-to-end network service 
requirements can be on connectivity, security, performance and fault-tolerance. However, there is a large 
conceptual gap between end-to-end requirements and detailed component configurations. To bridge this 
gap, a number of subsidiary requirements are created that constrain, for example, the protocols to be used, 
and the logical structures and associated policies to be set up at different protocol layers. By performing 
different types of reasoning with these requirements, different configuration tasks are accomplished. These 
include configuration synthesis, configuration error diagnosis, configuration error fixing, reconfiguration as 
requirements or components are added and deleted, and requirement verification. However, such reasoning 
is currently ad hoc. Network requirements are not even precisely specified hence automation of reasoning 
is impossible.  This is a major reason for the high cost of network management and total cost of 
ownership2. 
 
This paper proposes an approach for formalizing network requirements and automating associated 
reasoning. It is based on a new logical system called Alloy3. While Alloy is based in set theory, a subset of 
it also has an intuitive object-oriented interpretation: it lets one specify object types, their attributes and 
type of attribute values. It also lets one specify first-order logic constraints on these. Finally, it lets one 
specify a “scope” that defines a finite number of object instances of each type in a given system. Given a 
specification and a scope, Alloy attempts to find values of attributes of object instances in the scope that 
satisfy the specification. These values together constitute a “model” of the specification in the system in the 
logical sense of the word “model”. Alloy first compiles a specification into a propositional formula in 
conjunctive normal form, then uses a satisfiability solver such as Berkmin4 or Zchaff5 to check whether the 
formula is satisfiable. If so, it converts satisfying values of propositional variables back into values of 
attributes and displays these. Often, more than one solution is found.  
 

                                                 
2 ...operator error is the largest cause of failures...and largest contributor to time to repair ... in two of the three (surveyed) 
ISPs.......configuration errors are the largest category of operator errors. –  David Oppenheimer, Archana Ganapathi, David A. 
Patterson. Why Internet Services Fail and What Can Be Done About These? Proceedings of 4th Usenix Symposium on Internet 
Technologies and Systems (USITS ‘03), 2003. http://roc.cs.berkeley.edu/papers/usits03.pdf 
 
Although setup (of the trusted computing base) is much simpler than code, it is still complicated, it is usually done by less skilled 
people, and while code is written once, setup is different for every installation. So we should expect that it’s usually wrong, and many 
studies confirm this expectation. – Butler Lampson,  Computer Security In the Real World. Proceedings of Annual Computer Security 
Applications Conference, 2000. http://research.microsoft.com/lampson/64-SecurityInRealWorld/Acrobat.pdf 
 
Consider this: ….the complexity [of computer systems] is growing beyond human ability to manage it….the overlapping connections, 
dependencies, and interacting applications call for administrative decision-making and responses faster than any human can deliver. 
Pinpointing root causes of failures becomes more difficult. –Paul Horn, Senior VP, IBM Research. Autonomic Computing : IBM’s 
Perspective on the State of Information Technology. http://www.research.ibm.com/autonomic/manifesto/autonomic_computing.pdf 
 
3 Alloy. http://alloy.mit.edu 
4 Berkmin. http://eigold.tripod.com/BerkMin.html 
5 Zchaff. http://www.princeton.edu/~chaff/ 

Alloy is used to realize the concept of a Requirement Solver as shown in Figure 1. This takes as input two 
items: a set of network components, and requirements upon component configurations, and produces as 
output, component configurations satisfying those requirements.  

Component Configurations Satisfying Requirements

Requirements On
Component ConfigurationsSet of Network Components

Requirement
Solver

 
Figure 1. Requirement Solver 

 
The Solver has a direct implementation in Alloy. Network component types, attributes and values are 
directly modeled in Alloy. A set of network components of different types is modeled as a scope. Network 
system configuration is modeled as values of all component attributes in a given scope. Network 
requirements are modeled using first-order logic. Solutions are found by the Alloy model finder.  
 
The Solver can be used to accomplish a variety of reasoning tasks:  
 

1. Configuration Synthesis. To determine how to configure a set of components so they satisfy a 
system requirement R, submit the set of components and R to the solver and take the output.  

 
2. Requirement Strengthening. If a set of components satisfies a system requirement R but must 

now satisfy another requirement S, then to reconfigure components, submit the set of components 
and R∧S to the solver and take the output.  

  
3. Component Addition. If a new component is to be added to a set of components already 

satisfying requirement R, then to configure the new component and possibly reconfigure existing 
components,  submit the new set of components and R  to the Solver and take the output.  

 
4. Requirement Verification. To prove that it is impossible for an undesirable requirement U to be 

true when a set of components satisfies requirement R, submit the set of components and R∧U to 
the Solver. If the Solver cannot find a solution, the assertion is proved.  

 
5. Configuration Error Detection. To check whether configuration of a given set of components is 

consistent with a requirement R, represent the configuration as a set C of constraints each of the 
form P=V where P is a configuration parameter and V its value. Then, submit the set of 

components and  R∧C to the solver. If the solver cannot produce a solution, a configuration error 
is detected. 

 
6. Configuration Error Fixing. If the configuration of a given set of components is inconsistent 

with a requirement R, then submit the set of components and R to the Solver and find a new 
solution that is as “close” as possible to the current configuration. 

 
The Solver is illustrated in depth, by carrying out tasks 1-4 above in the context of a realistic fault-tolerant 
virtual private network with remote access. Modern satisfiability solvers solve millions of constraints in 
millions of variables in seconds. However, poor requirements can easily nullify such speeds. Approaches 
for writing efficient requirements are outlined.  
 
Section 1.1 contains a brief illustration of Alloy’s capabilities that we use. Section 2 describes the design of 
a fault-tolerant VPN with remote access, and the challenges of setting it up. Section 3 outlines a 
formalization of the design in Alloy. Sections 4,5,6,7 describe how to accomplish, respectively, tasks 1-4 
above. Section 8 outlines approaches for writing efficient requirements. Section 9 outlines relationship with 
previous work. Section 10 contains a summary, conclusions and directions for future research.  
 
1.1 Alloy By Example 
 
The following three lines declare three object types: router, subnet and interface. The last type 
has two attributes, the chassis (of type router) to which it belongs, and network (of type subnet) 
on which it is placed. These attributes model configuration parameters of an interface.  
 

sig router {} 
sig subnet{} 
sig interface {chassis: router, network: subnet} 

 

The predicate spec below defines a specification whose model we will try to find. It is a conjunction of 
three constraints. The first states that for every router x there is an interface y whose chassis is x, i.e., every 
router has at least one interface. The second states that no two non-equal interfaces on the same router are 
placed on the same subnet. The third states that there is 1 router, 2 subnets and 2 interfaces in our network. 
These are the components we want to configure.  
 

pred spec ()  
 {all x:router | some y:interface | y.chassis = x} 
 {no disj x1,x2:interface |  
  x1.chassis=x2.chassis && x1.network = x2.network} 
 {#router=1 && #subnet=2 && #interface=2} 

 
Now, the Alloy command:  
 

run spec for 1 router, 2 subnet, 2 interface 
 

produces the following model (values of configuration parameters):  
 

chassis : =  
 {interface_0 -> router_0, interface_1 -> router_0} 
network : =  
    {interface_0 -> subnet_1, interface_1 -> subnet_0} 

 
The first two lines state that the value of chassis for interface_0 is router_0 and the value of 
chassis for interface_1 is router_0. Similarly, for the last two lines. Alloy automatically creates 
instances of objects such as router_0, subnet_0, subnet_1. Note that Alloy did not place 
interface_0 and interface_1 on the same subnet due to the second constraint. On the other hand, 
if we remove that constraint, Alloy produces the following additional solution:  

chassis : =  
    {interface_0 -> router_0, interface_1 -> router_0} 
network : =  
    {interface_0 -> subnet_0, interface_1 -> subnet_0} 

 
in which both interface_0 and interface_1 are placed on the same subnet. 

2. Fault-Tolerant Virtual Private Network With Remote 
Access 
 
 
 

Spoke 
Router

Hub Router

Remote

All external interfaces such 
as these are in OSPF domain

Router

Host
Host

All internal interfaces and GRE 

Firewall policies
permit ESP and IKE 

packets

 
Figure 2. Fault-Tolerant Virtual Private Network With Remote Access 

 
Our top-level goal is to synthesize a fault-tolerant network that enables hosts, including mobile hosts, at 
geographically distributed sites to securely collaborate. A network design for achieving this goal is now 
outlined. When this is implemented via configuration, one obtains a network of the type shown in Figure 2.  
 
The existence of a wide-area network, represented by the WAN router in the figure, is assumed. Each site 
has a gateway router called a spoke router whose external (or public) interface is connected to the WAN 
and whose internal (or private) interface is connected to hosts and servers in the site. A routing protocol is 
run on the external interfaces of spoke and WAN routers to automatically compute routes between these 
interfaces. As traffic between hosts and servers on different sites is intended to be private, it cannot be 
allowed to flow directly over the wide area network. In order to secure it, one possibility is to set up a full-
mesh of IPSec tunnels between gateway routers. However, full-mesh is not scalable since the number of 
tunnels increases as the square of the number of sites: for the 200 sites expected in our domain, the number 
of tunnels would be nearly 20,000. A scalable alternative is a hub-and-spoke architecture as shown. A 
certain number of hub routers is set up. Each spoke router sets up an IPSec tunnel to each hub router. 
Traffic from one site to another goes via two tunnel hops, one from its spoke router to a hub router and 
another from the hub router to the destination site’s spoke router. The number of tunnels now only 
increases linearly with the number of sites.  
 
The problem, however, is that if a hub router fails, connectivity between sites is lost. This is because the 
source spoke router will continue to send traffic through the IPSec tunnel to the failed hub router. IPSec has 
no notion of fault-tolerance that will enable the source spoke router to redirect traffic via another hub 
router. Routing protocols such as RIP or OSPF accomplish precisely this kind of fault-tolerance, however 

they are incompatible with IPSec. They do not recognize an IPSec tunnel as directly connecting two routers 
since there can be multiple physical hops in between. The solution is to set up a new type of tunnel, called 
GRE, between each hub and spoke router. The purpose of GRE is to create the illusion to a routing protocol 
that two routers are directly connected, even when there are multiple physical hops in between. This is done 
by creating new GRE interfaces at each tunnel end point and making these belong to the same point-to-
point subnet. Now, if a hub router fails, a routing protocol will automatically direct traffic through another 
GRE tunnel to another hub router, and then to the destination. Each GRE tunnel is then secured via an 
IPSec tunnel. Thus, the required fault-tolerant virtual private network is set up. If two hub router failures 
are to be tolerated, then three hub routers are required. Then, the number of tunnels to be set up is just 600 
(number of hub routers * number of spoke routers) or 3% of nearly 20,000 in the full mesh case.  
 
This solution has a useful defense-in-depth feature: there are two separate routing domains, the external one 
and the overlay one. No routes are exchanged between these. Thus, even if an adversary compromises the 
WAN router, he cannot send a packet to a host. The WAN router does not even have a route to the host.  
 
In order to enable remote users to securely collaborate, a remote access server is set up in “parallel” with a 
spoke router. A remote user connected to the WAN sets up an L2TP tunnel between his host and the server. 
This tunnel gives the illusion to the host of being directly connected to the internal interface of the sites’ 
spoke router. Consequently, all traffic between the host and any other host or server on the VPN is also 
secured. Again, one has to ensure that two separate routing protocols run on the access server, one for the 
private side and one for the public.  
 
In order to realize the above design, the following types of configuration parameters need to be set: 
 

1. Addressing: Router interface address, type and subnet mask. 
2. IPSec: Tunnel end points, hash and encryption algorithms, tunnel modes, preshared keys and 

traffic filters. 
3. OSPF: Whether it is enabled at an interface, and OSPF area and type identifiers.  
4. GRE: Tunnel end points and physical end points supporting GRE tunnels. 
5. Firewall: Policies at each site.  
6. Remote access: Subnets to which remote access server interfaces belong and routing protocols 

enabled on these.   
 

It is very hard to compute values of the above configuration parameters. The types of configuration errors 
that can arise are: 
 

1. Duplicate IP addresses may be set up, or all interfaces on a subnet may not have the same type.  
2. IPSec tunnels may be set up incorrectly. For example, the preshared key, hash algorithm, 

encryption algorithm, or authentication mode may be unequal at the two tunnel end points. Peer 
values may not be mirror images of each other. These errors can lead to loss of connectivity. If the 
wrong traffic filter is used, then sensitive data can be transmitted without being encrypted.     

3. OSPF routing domain may be set up incorrectly, for example, it may not be enabled at a required 
interface or the area and type identifiers may be incorrect. This can lead to incorrect routing tables 
and to outright isolation of subnets.  

4. Routing loops may arise. If the same OSPF process is also used for routing between the gateway 
and WAN routers, then if it does not find a path through the physical network it will attempt to 
find a path through the overlay network. Since the overlay network is supported by the physical 
network, a routing loop will arise.  This problem can be mitigated by using two distinct routing 
protocols, one for the overlay and another for the WAN.  

5. GRE tunnels may be set up incorrectly. For example, the peer values may not be mirror images of 
each other, or the mapping between GRE ports and physical ports may be incorrect.  

6. Firewall policies may block IPSec traffic, hence no traffic will pass through the tunnels.  
7. Remote access interfaces may not belong to the correct subnets and incorrect routing protocols 

may be configured on these.  

Before we show how to formalize the above design in Alloy, we capture its main intuitions in the following 
requirements:  
 
RouterInterfaceRequirements 

1. Each spoke router has internal and external interfaces 
2. Each access server has internal and external interfaces  
3. Each hub router has only external interfaces 
4. Each WAN router has only external interfaces 

 
SubnettingRequirements 

5. A router does not have more than one interface on a subnet  
6. All internal interfaces are on internal subnets 
7. All external interfaces are on external subnets 
8. Every hub and spoke router is connected to a WAN router 
9. No two non-WAN routers share a subnet 

 
RoutingRequirements 

10. RIP is enabled on all internal interfaces 
11. OSPF is enabled on all external interfaces  

 
GRERequirements 

12. There is a GRE tunnel between each hub and spoke router  
13. RIP is enabled on all GRE interfaces 

 
SecureGRERequirements 

14. For every GRE tunnel there is an IPSec tunnel between associated physical interfaces that secures 
all GRE traffic 

 
AccessServerRequirements 

15. There exists an access server and spoke router such that the server is attached in “parallel” to the 
router 

 
AccessControlPolicyRequirements 

16. Each hub and spoke external interface permits esp and ike packets 
 
The interesting fact is these requirements do not specify the number of sites in the VPN. Rather, they apply 
to all sites. As new sites are added, these requirements are instantiated for the extended network to 
determine how to configure new components and reconfigure existing ones. This is a hard problem, in 
general, but that we show how to automatically solve with our approach.    

3. Requirement Formalization In Alloy 
 
This section presents an Alloy formalization of network component types, subtypes and their attributes.  
It also presents a formalization of Requirements 12 and 14. The complete formalization is provided in the 
Appendix.  
 
Various types of routers are modeled using the following Alloy type declarations (signatures):  

sig router {} 
sig wanRouter extends router {} 
sig hubRouter extends router {} 
sig spokeRouter extends router {} 
sig accessServer extends router {} 
sig legacyRouter extends router {} 

 
A generic interface just has a single attribute, the routing protocol enabled at it. 

sig interface {routing:routingDomain}  
 
A physical interface has two attributes, the router chassis on which it is mounted, and the network on which 
it is placed: 

sig physicalInterface extends interface { 
 chassis: router, 
 network: subnet} 

 
There are internal and external interfaces. External interfaces are of two types, one on hubs and one on 
spokes:  

sig internalInterface extends physicalInterface {} 
sig externalInterface extends physicalInterface {}  
sig hubExternalInterface extends externalInterface {} 
sig spokeExternalInterface extends externalInterface {}  

 
There are two types of routing domains, RIP and OSPF:  

sig routingDomain {} 
sig ripDomain extends routingDomain {} 
sig ospfDomain extends routingDomain {} 

 
There are two types of subnets, internal and external:  

sig subnet{} 
sig internalSubnet extends subnet{} 
sig externalSubnet extends subnet{} 

 
There are three types of protocols, IKE, ESP and GRE:  

sig protocol {} 
sig ike extends protocol {} 
sig esp extends protocol {} 
sig gre extends protocol {} 

 
A firewall policy contains one of two possible permissions, permit and deny, that respectively, mean 
whether the firewall should allow a packet to go through or be dropped.  

sig permission {} 
sig permit extends permission {} 
sig deny extends permission {} 

 
A firewall policy defines whether a packet associated with a protocol is allowed to go through or dropped, 
as it leaves an interface.   

sig FirewallPolicy { 
  prot: protocol, 
  action: permission,  
  protectedInterface: physicalInterface} 

An IPSec tunnel encrypts all packets associated with a protocol entering at either its local or its remote 
endpoint.     

sig ipsecTunnel { 
 local: externalInterface, 
 remote: externalInterface,  
 protocolToSecure: protocol} 

 
A GRE tunnel encapsulates a packet into a new packet with source address that of its local endpoint and 
destination address that of its remote endpoint. Also, the tunnel is considered a proper link in a routing 
domain.  

sig greTunnel { 
 localPhysical: externalInterface,  
 routing: routingDomain,   
 remotePhysical: externalInterface} 

 
An IP packet’s attributes are its source and destination interfaces and the protocol it embodies. The precise 
data it carries is not modeled, since it is not relevant for our design purposes.  

sig ipPacket { 
 source:interface, 
 destination:interface,  
 prot:protocol} 

 
The Alloy version of Requirement 12 is:  

{all x:hubExternalInterface, y:spokeExternalInterface | some g:greTunnel |  
      (g.localPhysical=x && g.remotePhysical=y) or 
      (g.localPhysical=y && g.remotePhysical=x)}  

 
This states that between every hubExternalInterface x and spokeExternalInterface y 
there is a greTunnel whose local physical is x and remotePhysical is y, or vice versa.   
 
The Alloy version of Requirement 14 is:  

{all g:greTunnel |  
 some p:ipsecTunnel | p.protocolToSecure=gre &&  
 ((p.local = g.localPhysical && p.remote = g.remotePhysical) or 
  (p.local = g.localPhysical && p.remote = g.remotePhysical))} 

 
This states that for every greTunnel g there is an ipsecTunnel p that secures the gre protocol 
and whose endpoints are the same as the physical endpoints of g. 

4. Configuration Synthesis 
 
This section shows how to synthesize the initial network with connectivity and routing. Define: 

PhysicalSpec =   
 RouterInterfaceRequirements ∧  
 SubnettingRequirements ∧  
 RoutingRequirements 
  

In Alloy, this would be expressed as:  
 

Pred PhysicalSpec () { 
 RouterInterfaceRequirements () 
 SubnettingRequirements () 
 RoutingRequirements ()} 

 
Define a scope consisting of: 

• 1 hubRouter 
• 1 spokeRouter  
• 1 wanRouter  
• 1 internalInterface 
• 4 externalInterface  
• 1 hubExternalInterface 
• 1 spokeExternalInterface 
• 1 ripDomain 
• 1 ospfDomain 
• 3 subnet 
• 0 legacyRouter 

 
These are the objects we want to configure. Now request Alloy to find a model for PhysicalSpec in the 
above scope. It synthesizes the network shown in Figure 3:  

Figure 3. Configuration Synthesis: Physical Network 

 
PhysicalSpec = (RouterInterfaceRequirements ∧ SubnettingRequirements ∧ 

RoutingRequirements) 
 

It does so by producing the following values of configuration parameters:  
 

routing : = 
    {externalInterface_0 -> ospfDomain_0,  
     externalInterface_1 -> ospfDomain_0,  
     hubExternalInterface_0 -> ospfDomain_0,  
     internalInterface_0 -> ripDomain_0,  
     spokeExternalInterface_0 -> ospfDomain_0} 
chassis : = 
    {externalInterface_0 -> wanRouter_0,  
     externalInterface_1 -> wanRouter_0,  
     hubExternalInterface_0 -> hubRouter_0,  
     internalInterface_0 -> spokeRouter_0,  
     spokeExternalInterface_0 -> spokeRouter_0} 
network : = 
    {externalInterface_0 -> externalSubnet_1,  
     externalInterface_1 -> externalSubnet_0,  
     hubExternalInterface_0 -> externalSubnet_0,  
     internalInterface_0 -> internalSubnet_0,  
     spokeExternalInterface_0 -> externalSubnet_1} 

 
These are just the textual version of the network in Figure 3. Note that Alloy automatically produces 
instances of object types in the scope, e.g., wanRouter_0, hubRouter_0, spokeRouter_0. Also 
note that spoke and hub routers are not directly connected, in accordance with Requirement 9. 

5. Requirement Strengthening 
 
In order to add an overlay network to the previous one, extend the previous scope with a GRE tunnel then 
request Alloy to satisfy (PhysicalSpec ∧ GRERequirements). Alloy synthesizes the following 
network: 
 
 

Spoke 
Router

Hub 
Router

OSPF Domain

RIP Domain
GRE Tunnel

WAN 
Router

 
Figure 4a. Requirement Strengthening: Adding Overlay 

(PhysicalSpec ∧ GRERequirements) 
 

Alloy automatically sets up the GRE tunnel between the spoke and hub router and enables RIP routing on 
the GRE tunnel.  
 
To make GRE tunnels secure, extend the previous scope with an IPSec tunnel and request Alloy to satisfy 
(PhysicalSpec ∧ GRERequirements ∧ SecureGRERequirements). Alloy synthesizes the 
following network:  

Figure 4b. Requirement Strengthening: Securing Overlay 

(PhysicalSpec ∧ GRERequirements ∧ SecureGRERequirements) 
 
Alloy automatically places the IPSec tunnel between the correct physical interfaces to protect the GRE 
tunnel.  
 
In order to add an access server to this network extend the previous scope with an access server, one 
internal interface, and one external interface and request Alloy to satisfy (PhysicalSpec ∧ 
GRERequirements ∧ SecureGRERequirements ∧ AccessServerRequirements). 
Alloy synthesizes the following network:  
 



 15

Spoke 
Router

Hub Router

Access Server

WAN 
Router

 
Figure 4c. Requirement Strengthening: Adding Remote Access Server 

 (PhysicalSpec ∧ GRERequirements ∧ SecureGRERequirements ∧ 
AccessServerRequirements) 

 
Note that the access server is placed in parallel with only the spoke router, not with any other router, and 
has the correct routing protocols enabled on its interfaces.  

6. Component Addition 
 
When new components are added to an infrastructure, the logic that governs infrastructure has to be 
instantiated to the extended set of components. This is a nontrivial problem for humans to cope with. With 
Alloy, this instantiation is accomplished simply by finding a model of requirements for the existing scope 
extended by new components. For example, in order to add a new spoke site to the previous network, 
extend its scope with a spoke router, one internal subnet, one external subnet, one GRE tunnel and one 
IPSec tunnel. Requesting Alloy to synthesize a network satisfying (PhysicalSpec ∧ 
GRERequirements ∧ SecureGRERequirements ∧ AccessServerRequirements)in the 
new scope yields the following network:  
 

Spoke 
Router

Spoke 
Router

Access Server

Hub Router

WAN 
Router

 
Figure 5a. Component Addition: Adding New Spoke Router 

(PhysicalSpec ∧ GRERequirements ∧ SecureGRERequirements ∧ 
AccessServerRequirements) 

 
Note that the new spoke router is physically connected just to the WAN router as required by Requirement 
8. Moreover, GRE and IPSec tunnels are automatically set up between the new spoke router and hub router 
and physical interfaces and GRE tunnels are placed in the correct routing domains.  
 
In order to add a new hub site to this network, extend its scope with a hub router, one external interface, 
one external subnet, two GRE tunnels and two IPSec tunnels. Requesting Alloy to synthesize a network 
satisfying (PhysicalSpec ∧ GRERequirements ∧ SecureGRERequirements ∧ 
AccessServerRequirements)in the new scope yields the following network:  

Figure 5b. Component Addition: Adding New Hub Router 
(PhysicalSpec ∧ GRERequirements ∧ SecureGRERequirements ∧ 

AccessServerRequirements) 
 

Finally, in order to permit IKE and ESP (protocols of IPSec) packets through the physical interfaces of hub 
and spoke routers, one can extend the above scope with 8 firewall policies, then request Alloy to satisfy:  
 
FullVPNSpec =  (PhysicalSpec ∧ GRERequirements ∧ SecureGRERequirements ∧ 
AccessServerRequirements ∧ FirewallPolicyRequirements)  
 
Alloy then synthesizes the network of Figure 2 without the hosts. The reason for 8 firewall policies is that 
one policy is required for each IPSec tunnel endpoint.  

7. Requirement Verification  
 
7.1  Identifying Incorrect Firewall Policies 
 
When we deployed the above network, we were careful to allow IKE and ESP packets to be permitted by 
access control lists at physical interfaces of hub and spoke routers. This was the reason for 
FirewallPolicyRequirements. However, we discovered that end-to-end connectivity was still not 
established. After considerable testing and analysis we realized that the WAN router itself was blocking 
IKE and ESP packets. We had not anticipated this cause.  
 
We now show how to formalize identification of this cause. Alloy was not used for such identification, but 
this example illustrates how it could have been. Define a condition called BlockedIPSec capturing 
conditions under which an IPSec packet can be blocked, and find out how it is possible that 
(FullVPNSpec ∧ BlockedIPSec) be true. In other words, is it possible that the network be 
configured in a manner consistent with FullVPNSpec yet block IPSec packets? If so, we would have to 
modify requirements to preclude this possibility.  
 
The predicate below states that IPSec is blocked if there is some esp or ike packet which is blocked.  

 
pred BlockedIPSec () 
 {some p:ipPacket, s,t:externalInterface |  
  p.source = s &&  
  p.destination = t &&  
  (p.prot = ike or p.prot=esp) &&  
  Blocked(p)} 

 
The predicate below states that a packet is blocked if there is some firewall policy protecting an external 
interface that denies the protocol for that packet: 
 

pred Blocked(pack:ipPacket) { 
  some p:firewallPolicy, x:externalInterface |  
  p.protectedInterface = x &&  
  p.prot=pack.prot &&  
  p.action = deny 
  } 
 

Now, if we increase the scope of the last network to include 9 (more than 8) firewall policies, and request 
Alloy to find a model for (FullVPNSpec ∧ BlockedIPSec), Alloy produces the following values 
for prot, permission and protectedInterface attributes of firewall policies:  
 

prot : =  
    {firewallPolicy_0 -> ike_0,  
     firewallPolicy_1 -> ike_0,  
     firewallPolicy_2 -> ike_0,  
     firewallPolicy_3 -> ike_0,  
     firewallPolicy_4 -> esp_0,  
     firewallPolicy_5 -> esp_0,  
     firewallPolicy_6 -> esp_0,  
     firewallPolicy_7 -> esp_0,  
     firewallPolicy_8 -> ike_0} 
 
permission: = 
    {firewallPolicy_0 -> permit_0,  
     firewallPolicy_1 -> permit_0,  
     firewallPolicy_2 -> permit_0,  
     firewallPolicy_3 -> permit_0,  
     firewallPolicy_4 -> permit_0,  
     firewallPolicy_5 -> permit_0,  
     firewallPolicy_6 -> permit_0,  
     firewallPolicy_7 -> permit_0,  

     firewallPolicy_8 -> deny_0} 
 
protectedInterface :  =  
    {firewallPolicy_0 -> spokeExternalInterface_1,  
     firewallPolicy_1 -> spokeExternalInterface_0,  
     firewallPolicy_2 -> hubExternalInterface_1,  
     firewallPolicy_3 -> hubExternalInterface_0,  
     firewallPolicy_4 -> spokeExternalInterface_1,  
     firewallPolicy_5 -> spokeExternalInterface_0,  
     firewallPolicy_6 -> hubExternalInterface_1,  
     firewallPolicy_7 -> hubExternalInterface_0,  
     firewallPolicy_8 -> externalInterface_0} 

 
In other words, firewallPolicy_8, applied on externalInterface_0 on the WAN router, 
blocks ike_0.  
 
7.2  Identifying Advertisement Of Private Subnets Into WAN 

Figure 6. Advertisement of internal subnet Y into WAN by legacy router 
 
This section illustrates another problem that arose during deployment of our VPN solution into an existing 
network. Existing networks contain “legacy” routers that have no concept of internal or external subnets as 
do spoke routers. Thus, if our VPN is grafted into an existing network as shown in the figure above, the 
defense-in-depth feature mentioned in Section 2 is compromised. The legacy router can run the same 
routing protocol on both its internal and external interfaces and thereby export the internal subnet Y to the 
WAN. Now, if the WAN router is compromised, an adversary can send packets to the host at Y. We now 
show how to formalize identification of this possibility. We define:  
 

pred internalSubnetAdvertisedToWan () 
 {some r:legacyRouter, x:physicalInterface, y:physicalInterface,  
  s:internalSubnet, e:externalSubnet |  
  x.chassis=r &&  
  y.chassis=r &&  

  x.network=s &&  
  y.network=e &&  
  x.routing=y.routing} 

 
This predicate states that an internal subnet is advertised to the WAN if there is a legacy router with two 
interfaces, one attached to an internal subnet and another to an external subnet, and both have the same 
routing protocol enabled on them. Now, if we increase the scope of the network of  Figure 5b to include 
one additional (legacy) router and two additional physical interfaces, and request Alloy to find a model for 
(FullVPNSpec ∧ internalSubnetAdvertisedToWan),Alloy produces the following:  
 

routing :  =  
    {physicalInterface_0 -> ospfDomain_0,  
     physicalInterface_1 -> ospfDomain_0,  
     ……} 
  
chassis :  =  
    {physicalInterface_0 -> legacyRouter_0,  
     physicalInterface_1 -> legacyRouter_0,  
     ……} 
  
network :  =  
    {  
     physicalInterface_0 -> externalSubnet_3,  
     physicalInterface_1 -> internalSubnet_1,  
     ……} 

 
In other words, PhysicalInterface_0 and physicalInterface_1 can be placed on 
legacyRouter_0, one can be connected to an internal subnet, the other to an external subnet, and yet 
both can belong to ospfDomain_0. 

8. Writing Efficient Requirements 
 
8.1 Scope Splitting 
 
One critical parameter to control in Alloy is the size of the scope. If it gets too large it should be split up 
and the specification changed, if necessary. Consider the following specification declaring router and 
interface types, and a relation chassis mapping an interface to its router. Also define EmptyCond 
to be an empty set of constraints to satisfy (Alloy requires some constraint before it can be run).  
 

sig router {} 
sig interface {chassis: router}  
pred EmptyCond () {} 

 
When Alloy tries to find a model for EmptyCond in a scope consisting of 50 routers and 50 interfaces it 
crashes! This is because the cross product of the set of all routers and chassis’ has 50*50=2500 pairs. Each 
subset of this product is a value of the chassis relation. Since there are 2^2500 subsets, there are that 
many possible values to enumerate.  
 
We can now try splitting the scope and redefining the specification:  
 

sig hubRouter {} 
sig spokeRouter {} 
sig hubRouterInterface {chassis:hubRouter} 
sig spokeRouterInterface {chassis:spokeRouter} 

 
Now, Alloy returns a model of EmptyCond for the scope consisting of 25 hubRouters, 25 
spokeRouters, 25 hubRouterInterfaces and 25 spokeRouterInterfaces in seconds! 
Note that the scope still contains 50 routers and 50 interfaces. But there are now “only” 2^625 * 2^625 = 
2^1250 possible values of chassis relation, or a factor of 2^1250 less.  
 
The scope splitting heuristic has been followed to structure the space of different routers and interfaces in 
the fault-tolerant VPN.  
 
8.2  Minimizing Number Of Quantifiers In Formulas 
 
Requirements containing quantifiers are transformed into Boolean form by instantiating quantified 
variables in all possible ways. The number of instantiations is the product of the number of instantiations of 
each quantified variable. The number of instantiations of each quantified variable is the size of the scope of 
that variable. In order to prevent the Boolean form from becoming excessively large, one can keep the 
number of quantified variables in a requirement as small as possible. For example, consider the definition 
of FirewallPolicyRequirements in which only two explicit quantifiers appear per requirement:  
 

pred FirewallPolicyRequirements ()  
{ 
(all t:ipsecTunnel | some p1:firewallPolicy |  
 p1.protectedInterface = t.local &&  
 p1.prot = ike && 
 p1.action = permit) && 
(all t:ipsecTunnel | some p1:firewallPolicy |  
 p1.protectedInterface =t.remote &&  
 p1.prot = ike && 
 p1.action = permit) && 
(all t:ipsecTunnel | some p1:firewallPolicy |  
 p1.protectedInterface = t.local &&  
 p1.prot = esp &&  
 p1.action = permit) &&  

(all t:ipsecTunnel | some p1:firewallPolicy |  
 p1.protectedInterface = t.remote &&  
 p1.prot = esp && 
 p1.action = permit)  
(no disj p1,p2:firewallPolicy | p1.protectedInterface=p2.protectedInterface &&  
 p1.prot=p2.prot && !p1.action=p2.action) 
} 

 
A more compact definition in which five quantifiers appear in a single requirement is:  
 

pred FirewallPolicyRequirements ()  
{ 
(all t:ipsecTunnel | some p1,p2,p3,p4:firewallPolicy |  
 p1.protectedInterface = t.local &&  
 p1.prot = ike && 
 p1.action = permit && 
 
 p2.protectedInterface = t.remote &&  
 p2.prot = ike &&  
 p2.action = permit &&  
 
 p3.protectedInterface = t.local &&  
 p3.prot = esp && 
 p3.action = permit && 
 
 p4.protectedInterface = t.remote &&  
 p4.prot = esp &&  
 p4.action = permit)&&  
 
(no disj p1,p2:firewallPolicy | p1.protectedInterface=p2.protectedInterface &&  
 p1.prot=p2.prot && !p1.action=p2.action) 
} 

In both cases, the number of IPSec tunnels in the scope is 4 and the number of firewall policies 9. However, 
there is a large difference in the size of the Boolean formula produced (for the entire VPN specification). In 
the first case, the formula contains 216,026 clauses and 73,4262 literals, and the entire process (compilation 
to solution) took 2 minutes and 59 seconds. In the second case, the formula contains 601,721 clauses and 
2,035,140 literals and the entire process took 8 minutes and 19 seconds. 
 
8.3 Avoiding Unintended Answers 
 
Consider the requirement:  
 

• Between every hub and spoke router there is a GRE tunnel  
 
This does not exclude GRE tunnels between hub and WAN routers, or between spoke and WAN routers or 
between WAN and WAN routers. We can strengthen the above requirement to preclude these possibilities, 
e.g.: 
 

• There is no GRE tunnel between a hub and WAN router 
• There is no GRE tunnel between a spoke and WAN router 
• There is no GRE tunnel between WAN routers 
• …. 

 
This can be quite unwieldy. An alternative is to specify just the first requirement but supply, in the scope, 
only the number of tunnels intended by the requirement. Alloy will consume these to satisfy the 
requirement. Thereafter, it will not have any more tunnels left to produce unintended possibilities. One will 
now have to carefully calculate the scope but at least requirements become much lesser in number. Not 
only is the size of associated Boolean formulas reduced, but also requirements become easier to visualize.  

9. Relationship With Previous Work 
 
IETF’s policy-based networking group6 has similar objectives as ours. Its main contributions are vendor-
neutral information models and if-condition-then-action rules called policies. Information models define 
types of objects, their attributes and possible values. These models, while important from a software 
development standpoint, are orthogonal to solving fundamental configuration management problems 
identified in this paper. These problems would remain even if we were to use all components from a single 
vendor. Policy-based networking also does not propose any declarative representation of system logic such 
as Requirements 1-16 of Section 2. Instead, if-condition-then-action rules are only procedural encodings of 
this logic. The problem with this approach is that these rules have to be changed or extended every time 
new requirements or components are added or deleted. In effect, these rules have to do all the work of the 
Requirement Solver (Alloy model finder). This is a formidable undertaking. Furthermore, verification with 
such procedural rules is impractical, and operations such as configuration error diagnosis or fixing are not 
addressed.  
 
In our approach, the Requirement Solver remains unchanged. It is only the requirements or the scope that 
change. The Requirement Solver automatically adjusts to these changes and finds new configurations. 
Verification is another application of the Requirement Solver.  
 
For the same reason, it is incorrect to assume that just the use of high-level languages like Perl and Python, 
often used in infrastructure management, can solve the fundamental configuration management problems. 
The hard part of reasoning from full first-order logic requirements still has to be programmed in these 
languages. It is this part that our approach automates.  
 
A previous paper7 formalized requirements of the type in Section 2, in Prolog. Prolog is based on definite 
clauses, hence it is not possible to use it to reason with full first-order logic constraints. Examples of these 
are “for every GRE tunnel there is an IPSec tunnel between associated physical interfaces that secures all 
GRE traffic” and “no two distinct interfaces on a router are on the same subnet”. The application of Prolog 
to system administration is thoroughly explored by Couch and Gilfix8. Related, widely used systems are 
Burgess’ CFEngine9 and Anderson’s LCFG10, but both have less expressive power than Prolog. These 
systems also perform robust application of configuration to components, a problem outside the scope of 
this paper. 
 
Recently, the need for specifying and reasoning with constraints on configurations has been amply 
expressed11. These constraints require the expressive power of full first-order logic, therefore our approach 
can address this need.  
 

                                                 
6 B. Moore, E. Ellesson, J. Strassner, A. Westerinen. Policy Core Information Model -- Version 1 Specification, IETF RFC 3060, 
February 2001. http://www.ietf.org/rfc/rfc3060.txt 
7 S. Narain, T. Cheng, B. Coan, V. Kaul, K. Parmeswaran, W. Stephens. Building Autonomic Systems Via Configuration. 
Proceedings of AMS Autonomic Computing Workshop, Seattle, WA, 2003. 
http://www.argreenhouse.com/papers/narain/Autonomic.pdf 
8 A. Couch, M. Gilfix. It’s Elementary, Dear Watson: Applying Logic Programming To Convergent System Management Processes. 
Proceedings of Large Installations Systems Administration Conference (LISA) 1999. 
http://www.eecs.tufts.edu/~mgilfix/publications/prolog-lisa99.pdf 
9 M. Burgess. A Site Configuration Engine. USENIX Computing systems, Vol8, No. 3, 1995. 
http://www.iu.hio.no/~mark/papers/paper1.pdf 
10 P. Anderson, A. Scobie. LCFG--The Next Generation. In Proceedings of UKUUG LISA/Winter Conference, 2002. 
http://www.lcfg.org/doc/ukuug2002.pdf.  
11 Configuration Workshop, Large Installations Systems Administration Conference (LISA) 2004. 
http://homepages.informatics.ed.ac.uk/group/lssconf/config2004/index.html 

An important configuration management problem, not discussed in this paper, is migration planning: in 
what order to configure components so that mission-critical invariants are never violated. For example, 
suppose the routing protocol on all routers has to be changed from RIP to OSPF. If the only method of 
accessing routers to perform this change is inband, then reconfiguring the first router to which the 
management station is attached will effectively isolate it from all others. This is because routing protocols 
compute routes to routers, but since OSPF and RIP processes do not exchange information with each other, 
the first router will not be able to compute routes to others.  The problem of the order in which to 
reconfigure components is fundamentally the problem of planning in artificial intelligence. The application 
of satisfiability solvers to this problem has been shown by Selman and Kautz12.  

 

                                                 
12 B. Selman, H. Kautz. Planning As Satisfiability. Proceedings of ECAI-92. 
http://www.cs.cornell.edu/selman/papers/pdf/92.ecai.satplan.pdf 

10. Summary, Conclusions & Future Directions 
 
This paper introduces the notion of a Requirement Solver and shows how fundamental configuration 
management tasks can be naturally formalized and carried out using it. These tasks are configuration 
synthesis, requirement strengthening, component addition, configuration error diagnosis and configuration 
error fixing. The Solver has been inspired by, and is implemented in, the new logical system called Alloy. 
Alloy is based on the concept of model finding for full first-order logic in finite domains, rather than 
theorem proving. The Solver is illustrated in the context of a realistic fault-tolerant VPN with remote 
access. Approaches for writing efficient specifications are outlined. 
 
Alloy’s strength is efficiently sorting through complex, full first-order logic constraints, provided scopes 
are small. Based on experiments of this paper, it is possible that Alloy be used from a traditional 
programming language to develop a new class of configuration management algorithms for networks of 
realistic scale and complexity. The heuristics of Section 8 could be programmed in the programming 
language. Other approaches for scalability are tuning satisfiability solvers to the networking domain, and 
improving Alloy compilers.  
 
It is not obvious how to select the least cost change from the current configuration as required for 
Configuration Error Fixing. It is also not obvious how to do incremental configuration for Requirement 
Strengthening and Component Addition. One shouldn’t have to recompute configurations for all 
components when a small number of new requirements or components are added. Finally, other approaches 
to writing efficient specifications need to be explored.  

Urbana Champaign for very useful ideas and feedback.  
 



 26

 

Appendix: Full Alloy Specification of Fault-Tolerant VPN 
of Figure 1 

 
module samples/router 
 
---------Routing protocol signatures 
sig routingDomain {} 
sig ripDomain extends routingDomain {} 
sig ospfDomain extends routingDomain {} 
 
----------Subnet signatures 
sig subnet{} 
sig internalSubnet extends subnet{} 
sig externalSubnet extends subnet{} 
 
----------Router signatures 
sig router {} 
sig wanRouter extends router {} 
sig hubRouter extends router {} 
sig spokeRouter extends router {} 
sig legacyRouter extends router {} 
sig accessServer extends router {} 
 
----------Interface signatures 
sig interface { 
 routing:routingDomain} 
sig physicalInterface extends interface { 
 chassis: router, 
 network: subnet} 
 
sig internalInterface extends physicalInterface {} 
sig externalInterface extends physicalInterface {}  
 
sig hubExternalInterface extends externalInterface {} 
sig spokeExternalInterface extends externalInterface {}  
 
----------Protocol signatures 
sig protocol {} 
sig ike extends protocol {} 
sig esp extends protocol {} 
sig gre extends protocol {} 
 
-----------Permission signatures 
sig permission {} 
sig permit extends permission {} 
sig deny extends permission {} 
 
----------Firewall policy signature 
sig firewallPolicy { 
  prot: protocol, 
  action: permission,  
  protectedInterface: physicalInterface} 
 
-----------IPSec Tunnel signature  
sig ipsecTunnel { 
 local: externalInterface, 
 remote: externalInterface,  
 protocolToSecure: protocol} 
 
----------GRE Tunnel signature 
sig greTunnel { 
 localPhysical: externalInterface,  
 routing: routingDomain,   
 remotePhysical: externalInterface} 

-----------IP Packet signature 
sig ipPacket { 
 source:interface, 
 destination:interface,  
 prot:protocol} 
 
-------------------------------------------------------------------- 
-- Each spoke router and access server have internal and external interfaces 
-- Each hub and wan router has external interfaces 
-------------------------------------------------------------------- 
 
pred RouterInterfaceRequirements ()  
{ 
 (all x:spokeRouter | some y:internalInterface | y.chassis = x) &&  
 (all x:spokeRouter | some y:spokeExternalInterface | y.chassis = x) &&  
 (all x:accessServer | some y:internalInterface | y.chassis = x) &&  
 (all x:accessServer | some y:externalInterface | y.chassis = x) &&  
 (all x:hubRouter | some y:hubExternalInterface | y.chassis = x)&&  
 (all x:wanRouter | some y:externalInterface | y.chassis = x) 
} 
 
---------------------------------------------------------------------- 
-- Setting up routing 
---------------------------------------------------------------------- 
pred RoutingRequirements () { 
 ripOnInternalInterfaces () 
 ospfOnExternalInterfaces ()} 
 
pred ripOnInternalInterfaces () { 
 all x:internalInterface | x.routing = ripDomain} 
 
pred ospfOnExternalInterfaces () { 
 all x:externalInterface | x.routing = ospfDomain} 
 
--------------------------------------------------------------------- 
-- Setting up subnetting 
-- A router does not have more than one interface on a subnet 
-- Also, internal interfaces are only placed on internal subnets.  
-- Same for external interfaces and subnets 
--------------------------------------------------------------------- 
 
pred SubnettingRequirements ()  
 { 
 routerInterfacesDistinctSubnets () 
 internalInterfaceSubnet () 
 externalInterfaceSubnet ()  
 hubSpokeConnectedToWan () 
 noNonWanRoutersDirectlyConnected () 
} 
 
pred routerInterfacesDistinctSubnets ()  
 {no disj x1,x2:physicalInterface |  
 x1.chassis=x2.chassis &&  
 x1.network = x2.network} 
 
pred internalInterfaceSubnet () 
 {all x:internalInterface | some s:internalSubnet | x.network = s} 
 
pred externalInterfaceSubnet ()  
 {all x:externalInterface | some s:externalSubnet | x.network = s} 
 
pred hubSpokeConnectedToWan ()  
{ spokeConnectedToWan () && 
  hubConnectedToWan ()} 
 
pred spokeConnectedToWan () 
{all x:spokeRouter | some w:wanRouter, u,v:externalInterface |  
 u.chassis=x && v.chassis=w && u.network=v.network} 
 
pred hubConnectedToWan () 
{all x:hubRouter | some w:wanRouter, u,v:externalInterface |  

 u.chassis=x && v.chassis=w && u.network=v.network} 
 
-------------------------------------------------------------------- 
-- No two hub routers share a subnet. 
-- No hub and spoke routers share a subnet 
-- No two spoke routers share a subnet 
-- Every hub and spoke router must be connected to a wan router 
-------------------------------------------------------------------- 
pred noNonWanRoutersDirectlyConnected ()  
{ 
(no disj x,y:externalInterface, disj s,t:spokeRouter |  
 x.chassis = s && y.chassis=t && x.network=y.network) 
(no disj x,y:externalInterface, s:spokeRouter, t:hubRouter |  
 x.chassis = s && y.chassis=t && x.network=y.network) 
(no disj x,y:externalInterface, disj s,t:hubRouter |  
 x.chassis = s && y.chassis=t && x.network=y.network) 
} 
------------------------------------------------------------------- 
-- Access server. Associated with some spoke router is an  
-- access server whose internal interfaces are on the same  
-- subnet as are their external ones.  
------------------------------------------------------------------- 
 
pred AccessServerRequirements () 
{some x:spokeRouter,  
      acc:accessServer,  
      xInt, accInt:internalInterface,  
      xExt, accExt:externalInterface 
 | xInt.chassis = x &&  
   accInt.chassis = acc &&  
   xInt.network = accInt.network && 
   accExt.chassis = acc &&  
   xExt.network = accExt.network} 
 
------------------------------------------------------------------- 
-- There is a GRE tunnel between every hub and spoke router 
------------------------------------------------------------------- 
 
pred GRERequirements () 
{greTunnelEveryHubSpoke () &&  
 ripOnGREInterfaces ()} 
  
pred greTunnelEveryHubSpoke () 
{all x:hubExternalInterface, y:spokeExternalInterface | some g:greTunnel |  
      (g.localPhysical=x && g.remotePhysical=y) or 
      (g.localPhysical=y && g.remotePhysical=x)}  
 
pred ripOnGREInterfaces () { 
 all g:greTunnel | g.routing = ripDomain} 
 
------------------------------------------------------------------- 
-- Securing GRE tunnels. Associated with each GRE tunnel is 
-- an IPSec tunnel between associated physical interfaces 
------------------------------------------------------------------- 
pred SecureGRERequirements ()  
{all g:greTunnel |  
 some p:ipsecTunnel | p.protocolToSecure=gre &&  
 ((p.local = g.localPhysical && p.remote = g.remotePhysical) or 
  (p.local = g.localPhysical && p.remote = g.remotePhysical))} 
 
---------------------------------------------------------------------- 
-- Firewall placement and policies 
-- Each IPSec tunnel is allowed through the hub and spoke router firewalls 
-- Also, there cannot be two contradictory policies associated with an interface 
---------------------------------------------------------------------- 
 
pred FirewallPolicyRequirements ()  
{ 
(all t:ipsecTunnel | some p1:firewallPolicy |  
 p1.protectedInterface = t.local &&  
 p1.prot = ike && 

 p1.action = permit) && 
 
(all t:ipsecTunnel | some p1:firewallPolicy |  
 p1.protectedInterface =t.remote &&  
 p1.prot = ike && 
 p1.action = permit) && 
 
(all t:ipsecTunnel | some p1:firewallPolicy |  
 p1.protectedInterface = t.local &&  
 p1.prot = esp &&  
 p1.action = permit) &&  
 
(all t:ipsecTunnel | some p1:firewallPolicy |  
 p1.protectedInterface = t.remote &&  
 p1.prot = esp && 
 p1.action = permit)  
 
(no disj p1,p2:firewallPolicy | p1.protectedInterface=p2.protectedInterface &&  
 p1.prot=p2.prot && !p1.action=p2.action) 
} 
--------------------------------------------------------------------- 
-- Figure 1  
--------------------------------------------------------------------- 
pred FullVPNSpec ()  
{ VPNScope () && 
 RouterInterfaceRequirements () &&  
 SubnettingRequirements () &&  
 RoutingRequirements () &&  
 GRERequirements () &&  
 SecureGRERequirements () &&  
 AccessServerRequirements () &&  
 FirewallPolicyRequirements ()  
} 
 
pred VPNScope () 
{ 
 #hubRouter=2 
 #spokeRouter = 2 
 #accessServer = 1 
 #wanRouter = 1 
 #internalInterface = 3 -- #spokeRouter + #accessServer 
 #externalInterface = 9 -- 2*(#hubRouter + #spokeRouter) + #accessServer 
 #hubExternalInterface = #hubRouter 
 #spokeExternalInterface = #spokeRouter 
 
-- #greTunnel = 4 -- #hubRouter * #spokeRouter 
-- #ipsecTunnel = 4 -- same as #greTunnel 
-- #firewallPolicy = 8 -- 2*(#hubRouter + #spokeRouter) 
-- #router = #hubRouter+#spokeRouter+#wanRouter+#legacyRouter+#accessServer 
-- #routingDomain = 2 
-- #ripDomain = 1  
-- #ospfDomain = 1 
} 
 
run FullVPNSpec for 12 interface, 4 greTunnel, 6 router, 3 protocol, 8 firewallPolicy, 4 
ipsecTunnel, 2 routingDomain, 1 ipPacket, 6 subnet, 2 permission 
 
-- run VerifyFirewall for 12 interface, 4 greTunnel, 6 router, 3 protocol, 9 
firewallPolicy, 4 ipsecTunnel, 2 routingDomain, 1 ipPacket, 6 subnet, 2 permission 
 
-- run VerifyInternalSubnetExport for 14 interface, 4 greTunnel, 7 router, 3 protocol, 8 
firewallPolicy, 4 ipsecTunnel, 2 routingDomain, 1 ipPacket, 6 subnet, 2 permission 
 
--------------------------------------------------------------------- 
-- Figure 2 
--------------------------------------------------------------------- 
pred PhysicalSpec ()  
{ PhysicalSpecScope () && 
 RouterInterfaceRequirements () &&  
 SubnettingRequirements () &&  
 RoutingRequirements () 

} 
 
pred PhysicalSpecScope () 
{ 
 #hubRouter=1 
 #spokeRouter = 1 
 #wanRouter = 1 
 
 #internalInterface = 1 
 #externalInterface = 4 -- 2*(#hubRouter + #spokeRouter) 
 #hubExternalInterface = #hubRouter 
 #spokeExternalInterface = #spokeRouter 
 
 #routingDomain = 2 
 #ripDomain = 1  
 #ospfDomain = 1 
} 
 
-- run PhysicalSpec for 5 interface, 0 greTunnel, 3 router, 0 protocol, 0 firewallPolicy, 
0 ipsecTunnel, 2 routingDomain, 0 ipPacket, 3 subnet, 0 permission 
 
 
/* 
-------------------------------------------------------------------- 
Verification Predicates 
-------------------------------------------------------------------- 
*/ 
 
pred VerifyFirewall () { 
 FullVPNSpec() 
 BlockedIPSec ()} 
 
pred VerifyInternalSubnetExport () { 
 FullVPNSpec ()  
 internalSubnetAdvertisedToWan ()} 
 
-------------------------------------------------------------------- 
-- IPSec is blocked if there is some esp or ike packet which is 
-- blocked.  
 
pred BlockedIPSec () 
 {some p:ipPacket, s,t:externalInterface |  
  p.source = s &&  
  p.destination = t &&  
  (p.prot = ike or p.prot=esp) &&  
  Blocked(p)} 
 
-- A packet is blocked if there is some firewall policy protecting an  
-- external interface that denies the protocol for that packet.  
 
pred Blocked(pack:ipPacket) { 
 some p:firewallPolicy, x:externalInterface |  
  p.protectedInterface = x &&  
  p.prot=pack.prot &&  
  p.action = deny 
  } 
-- A packet is tunneled if its source and destination are internal  
-- or GRE interfaces. It will be sent through the GRE/IPSec tunnel 
pred tunneled(p:ipPacket) { 
 (some s,t:internalInterface | p.source=s && p.destination=t) or  
  p.prot = gre} 
 
 
---------------------------------------------------------------------- 
-- Unsafe condition: internal subnet advertised to WAN 
---------------------------------------------------------------------- 
pred internalSubnetAdvertisedToWan () 
 {some r:legacyRouter, x:physicalInterface, y:physicalInterface,  
  s:internalSubnet, e:externalSubnet |  
  x.chassis=r &&  
  y.chassis=r &&  

  x.network=s &&  
  y.network=e &&  
  x.routing=y.routing} 
 
 
The Alloy command: 
 
run FullVPNSpec for 12 interface, 4 greTunnel, 6 router, 3 protocol, 8  
 firewallPolicy, 4 ipsecTunnel, 2 routingDomain, 1 ipPacket, 6 subnet, 2 permission 
 
returns the following:  
 
module samples/router 
sig routingDomain extends univ = {ospfDomain_0, ripDomain_0} 
sig ripDomain extends routingDomain = {ripDomain_0} 
sig ospfDomain extends routingDomain = {ospfDomain_0} 
sig subnet extends univ = {externalSubnet_0, externalSubnet_1, externalSubnet_2, 
externalSubnet_3, internalSubnet_0, subnet_0} 
sig internalSubnet extends subnet = {internalSubnet_0} 
sig externalSubnet extends subnet = {externalSubnet_0, externalSubnet_1, 
externalSubnet_2, externalSubnet_3} 
sig router extends univ = {accessServer_0, hubRouter_0, hubRouter_1, spokeRouter_0, 
spokeRouter_1, wanRouter_0} 
sig wanRouter extends router = {wanRouter_0} 
sig hubRouter extends router = {hubRouter_0, hubRouter_1} 
sig spokeRouter extends router = {spokeRouter_0, spokeRouter_1} 
sig legacyRouter extends router = {} 
sig accessServer extends router = {accessServer_0} 
sig interface extends univ = {externalInterface_0, externalInterface_1, 
externalInterface_2, externalInterface_3, externalInterface_4, hubExternalInterface_0, 
hubExternalInterface_1, internalInterface_0, internalInterface_1, internalInterface_2, 
spokeExternalInterface_0, spokeExternalInterface_1} 
  routing :  samples/router/routingDomain =  
    {externalInterface_0 -> ospfDomain_0,  
     externalInterface_1 -> ospfDomain_0,  
     externalInterface_2 -> ospfDomain_0,  
     externalInterface_3 -> ospfDomain_0,  
     externalInterface_4 -> ospfDomain_0,  
     hubExternalInterface_0 -> ospfDomain_0,  
     hubExternalInterface_1 -> ospfDomain_0,  
     internalInterface_0 -> ripDomain_0,  
     internalInterface_1 -> ripDomain_0,  
     internalInterface_2 -> ripDomain_0,  
     spokeExternalInterface_0 -> ospfDomain_0,  
     spokeExternalInterface_1 -> ospfDomain_0} 
sig physicalInterface extends interface = {externalInterface_0, externalInterface_1, 
externalInterface_2, externalInterface_3, externalInterface_4, hubExternalInterface_0, 
hubExternalInterface_1, internalInterface_0, internalInterface_1, internalInterface_2, 
spokeExternalInterface_0, spokeExternalInterface_1} 
  chassis :  samples/router/router =  
    {externalInterface_0 -> wanRouter_0,  
     externalInterface_1 -> wanRouter_0,  
     externalInterface_2 -> wanRouter_0,  
     externalInterface_3 -> accessServer_0,  
     externalInterface_4 -> wanRouter_0,  
     hubExternalInterface_0 -> hubRouter_1,  
     hubExternalInterface_1 -> hubRouter_0,  
     internalInterface_0 -> spokeRouter_0,  
     internalInterface_1 -> accessServer_0,  
     internalInterface_2 -> spokeRouter_1,  
     spokeExternalInterface_0 -> spokeRouter_1,  
     spokeExternalInterface_1 -> spokeRouter_0} 
  network :  samples/router/subnet =  
    {externalInterface_0 -> externalSubnet_3,  
     externalInterface_1 -> externalSubnet_0,  
     externalInterface_2 -> externalSubnet_1,  
     externalInterface_3 -> externalSubnet_2,  
     externalInterface_4 -> externalSubnet_2,  
     hubExternalInterface_0 -> externalSubnet_3,  
     hubExternalInterface_1 -> externalSubnet_2,  
     internalInterface_0 -> internalSubnet_0,  

     internalInterface_1 -> internalSubnet_0,  
     internalInterface_2 -> internalSubnet_0,  
     spokeExternalInterface_0 -> externalSubnet_1,  
     spokeExternalInterface_1 -> externalSubnet_0} 
sig internalInterface extends physicalInterface = {internalInterface_0, 
internalInterface_1, internalInterface_2} 
sig externalInterface extends physicalInterface = {externalInterface_0, 
externalInterface_1, externalInterface_2, externalInterface_3, externalInterface_4, 
hubExternalInterface_0, hubExternalInterface_1, spokeExternalInterface_0, 
spokeExternalInterface_1} 
 
sig hubExternalInterface extends externalInterface = {hubExternalInterface_0, 
hubExternalInterface_1} 
sig spokeExternalInterface extends externalInterface = {spokeExternalInterface_0, 
spokeExternalInterface_1} 
sig protocol extends univ = {esp_0, gre_0, ike_0} 
sig ike extends protocol = {ike_0} 
sig esp extends protocol = {esp_0} 
sig gre extends protocol = {gre_0} 
sig permission extends univ = {deny_0, permit_0} 
sig permit extends permission = {permit_0} 
sig deny extends permission = {deny_0} 
sig firewallPolicy extends univ = {firewallPolicy_0, firewallPolicy_1, firewallPolicy_2, 
firewallPolicy_3, firewallPolicy_4, firewallPolicy_5, firewallPolicy_6, firewallPolicy_7, 
firewallPolicy_8} 
  prot :  samples/router/protocol =  
    {firewallPolicy_0 -> ike_0,  
     firewallPolicy_1 -> ike_0,  
     firewallPolicy_2 -> ike_0,  
     firewallPolicy_3 -> ike_0,  
     firewallPolicy_4 -> esp_0,  
     firewallPolicy_5 -> esp_0,  
     firewallPolicy_6 -> esp_0,  
     firewallPolicy_7 -> esp_0,  
     firewallPolicy_8 -> esp_0} 
  action :  samples/router/permission =  
    {firewallPolicy_0 -> permit_0,  
     firewallPolicy_1 -> permit_0,  
     firewallPolicy_2 -> permit_0,  
     firewallPolicy_3 -> permit_0,  
     firewallPolicy_4 -> permit_0,  
     firewallPolicy_5 -> permit_0,  
     firewallPolicy_6 -> permit_0,  
     firewallPolicy_7 -> permit_0,  
     firewallPolicy_8 -> deny_0} 
  protectedInterface :  samples/router/physicalInterface =  
    {firewallPolicy_0 -> spokeExternalInterface_1,  
     firewallPolicy_1 -> spokeExternalInterface_0,  
     firewallPolicy_2 -> hubExternalInterface_1,  
     firewallPolicy_3 -> hubExternalInterface_0,  
     firewallPolicy_4 -> spokeExternalInterface_1,  
     firewallPolicy_5 -> spokeExternalInterface_0,  
     firewallPolicy_6 -> hubExternalInterface_1,  
     firewallPolicy_7 -> hubExternalInterface_0,  
     firewallPolicy_8 -> externalInterface_0} 
sig ipsecTunnel extends univ = {ipsecTunnel_0, ipsecTunnel_1, ipsecTunnel_2, 
ipsecTunnel_3} 
  local :  samples/router/externalInterface =  
    {ipsecTunnel_0 -> spokeExternalInterface_1,  
     ipsecTunnel_1 -> spokeExternalInterface_0,  
     ipsecTunnel_2 -> spokeExternalInterface_0,  
     ipsecTunnel_3 -> hubExternalInterface_0} 
  remote :  samples/router/externalInterface =  
    {ipsecTunnel_0 -> hubExternalInterface_1,  
     ipsecTunnel_1 -> hubExternalInterface_1,  
     ipsecTunnel_2 -> hubExternalInterface_0,  
     ipsecTunnel_3 -> spokeExternalInterface_1} 
  protocolToSecure :  samples/router/protocol =  
    {ipsecTunnel_0 -> gre_0,  
     ipsecTunnel_1 -> gre_0,  
     ipsecTunnel_2 -> gre_0,  

     ipsecTunnel_3 -> gre_0} 
sig greTunnel extends univ = {greTunnel_0, greTunnel_1, greTunnel_2, greTunnel_3} 
  localPhysical :  samples/router/externalInterface =  
    {greTunnel_0 -> spokeExternalInterface_1,  
     greTunnel_1 -> spokeExternalInterface_0,  
     greTunnel_2 -> spokeExternalInterface_0,  
     greTunnel_3 -> hubExternalInterface_0} 
  routing :  samples/router/routingDomain =  
    {greTunnel_0 -> ripDomain_0,  
     greTunnel_1 -> ripDomain_0,  
     greTunnel_2 -> ripDomain_0,  
     greTunnel_3 -> ripDomain_0} 
  remotePhysical :  samples/router/externalInterface =  
    {greTunnel_0 -> hubExternalInterface_1,  
     greTunnel_1 -> hubExternalInterface_1,  
     greTunnel_2 -> hubExternalInterface_0,  
     greTunnel_3 -> spokeExternalInterface_1} 
sig ipPacket extends univ = {ipPacket_0} 
  source :  samples/router/interface =  
    {ipPacket_0 -> externalInterface_2} 
  destination :  samples/router/interface =  
    {ipPacket_0 -> externalInterface_1} 
  prot :  samples/router/protocol =  
    {ipPacket_0 -> esp_0} 
 


